

リーマン面内の楕円柱まわり仮想的な流れのシミュレーション Simulation of virtual flow around an elliptical column in the Riemann surface

河村 哲也

抄録：本研究では、どのような状況も作り出せるという数値シミュレーションの利点を活かして、リーマン面内の流体の仮想的な流れのシミュレーションを行った。具体的には2枚のリーマン面と3枚のリーマン面内に楕円柱が置かれて、それに一様流があたっている状態を想定して流れのシミュレーションを行った。基礎方程式として非圧縮性のナビエ・ストークス方程式を用い、標準的なMAC法を用いて方程式を数値的に解いた。レイノルズ数は、カルマン渦が発生する200として、現実におきる流れと比較した。リーマン面が2枚の場合、2枚の面で一様流の流れをわずかに変化させたところ、それに応じてカルマン渦の位相がわずかにずれた。3枚の場合ではリーマン面によって双子渦と位相がほぼ180°ずれたカルマン渦の流れが現れた。

Abstract: In this study, the virtual flow of fluid in the Riemann surface is simulated by taking advantage of the numerical simulation that can create any situation. Specifically, it is assumed that one elliptical column is placed in two Riemann surfaces and three Riemann surfaces, and a uniform flow is applied to them. The incompressible Navier-Stokes equation is used as the basic equation, and the equation is solved numerically using the standard MAC method. The Reynolds number is set to 200, where Karman vortices are generated and the result is compared with the actual flow. When there are two Riemann surfaces, when the flow of uniform flow is slightly changed on the two surfaces, the phase of the Karman vortex is slightly shifted accordingly. In the case of three Riemann surfaces, the twin vortex flow and two kind of Karman vortex flow that is approximately 180 ° out of phase appeared on each surface.

キーワード：リーマン面、楕円柱、数値シミュレーション、流れ場、非圧縮性流れ

Key words : Riemann surface, Elliptical column, Numerical simulation, Flow field, Incompressible flow

1. はじめに

粘性流体の運動を記述する基礎方程式はナビエ・ストークス方程式とよばれ、19世紀半ばにすでに確立された。ナビエ・ストークス方程式を与えられた初期条件と境界条件のもとで解くことにより、流れ（速度場や圧力場）が完全にわかる。しかし、ナビエ・ストークス方程式は非線形の連立偏微分方程式であり、解析的に厳密に解くことは、領域形状により非線形項が現れないような、ごく特殊な場合を除いて不可能である。また、非線形性のため、たとえ厳密解が得られたとしても、それを重ね合わせて条件を満たす解を構成することもできない。しかし、1960年代に普及が始まったコンピュータは流体力学の分野でも、研究方法を一変させた。すなわち、基礎方程式を数値的に、近似的に解くことによって解を調べることができるようになった。このような手法は数値シミュレーションとよばれている。そして、コンピュータの長足の進歩は日常生活に現れる複雑な流れに対しても、数値シミュレーションにより十分に精度がよい解を得ることを可能とし、数値シミュレーションが実験にとってかわりつつある。

数値シミュレーションはこのように実用的に非常に重要な役割を果たすとともに、どのような状況もコンピュータ内で再現することができるため、現実には起こり得ない仮想的な状況も簡単に作り出すことができる。その一例として4次元物体まわりの4次元流れがある¹。たとえ非現実であっても、次元を上げて流れを考察することにより、現実の3次元流れの理解に対して新たな視点を付け加えることができる。本論文では、別の仮想的な流れとして、リーマン面内の楕円柱まわりの流れを取り上げる。

2. リーマン面

まず、複素関数

$$w = z^{1/2} \quad (1)$$

を考える。 z を極形式

$$z = r e^{i\theta} = r e^{i\theta + 2n\pi i} \quad (n: \text{整数}, 0 \leq \theta < 2\pi) \quad (2)$$

で表し、式(1)に代入すると

$$w = r^{1/2} e^{i\theta/2} e^{in\pi}$$

すなわち

$$w = r^{1/2} e^{i\theta/2} \quad (n: \text{偶数}), \quad w = -r^{1/2} e^{i\theta/2} \quad (n: \text{奇数})$$

となる。このことは z 面で1点を表す $z = r e^{i(\theta + 2n\pi)}$ が、 w 面では n が偶数か奇数かに応じて、上半平面にある点または下半平面にある点に写像されることを意味している。したがって、 $w = z^{1/2}$ は1つの z に対応して2つの値をとる2価関数になっている(図1)。それゆえ、 w 面全体を z 面で表現するには2枚の z 面が必要になる。1枚は、 $2n\pi$ の不定性を除いて $0 < \theta < 2\pi$ で、もう1枚は $2\pi < \theta < 4\pi$ である。




図1 2価関数




図2 $w = z^{1/2}$ に対するリーマン面

この2つの面をたとえば図2に示すように実軸の正の部分でつながるようにする。このとき原点まわりの閉曲線を考えると、切れ目から1枚目を通して1周すると2枚目に入り、2枚目を1周すると1枚目に戻ることになる。このようにしてつくった面を、 $(z^{1/2})$ に対するリーマン面という。要点は、原点を2周してはじめて同じ点に戻ると考えた点で、このようにすれば関数の2価性を解消することができる。切れ目は原点を出発して無限遠に到達する自分自身は交わらな

い曲線であれば任意に選ぶことができ、この切れ目を横切るときもう1方の面に入ると約束する。同様に考えれば、 $w=z^{1/3}$ に対するリーマン面は3枚ある。本研究で取り扱うのはこのような2枚または3枚のリーマン面内に楕円柱があるときの流れである。

3. 計算法

2次元非圧縮性流れは

$$\frac{\partial u}{\partial x} + \frac{\partial v}{\partial y} = 0 \quad (3)$$

$$\frac{\partial u}{\partial t} + u \frac{\partial u}{\partial x} + v \frac{\partial u}{\partial y} = -\frac{\partial p}{\partial x} + \frac{1}{Re} \left(\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} \right) \quad (4)$$

$$\frac{\partial v}{\partial t} + u \frac{\partial v}{\partial x} + v \frac{\partial v}{\partial y} = -\frac{\partial p}{\partial y} + \frac{1}{Re} \left(\frac{\partial^2 v}{\partial x^2} + \frac{\partial^2 v}{\partial y^2} \right) \quad (5)$$

によって支配される。(u,v)は無次元化された速度の(x,y)成分、pは無次元化された圧力、Reはレイノルズ数である。これらの方程式を数値的に解く標準的な方法にMAC法がある。MAC法では式(2)、(3)から式(1)を考慮して得られる圧力に対する次のポアソン方程式を解いて圧力を求める。

$$\frac{\partial^2 p}{\partial x^2} + \frac{\partial^2 p}{\partial y^2} = \left(\frac{\partial u}{\partial x} \right)^2 + 2 \frac{\partial u}{\partial y} \frac{\partial v}{\partial x} + \left(\frac{\partial v}{\partial y} \right)^2 + \frac{1}{\Delta t} \left(\frac{\partial u}{\partial x} + \frac{\partial v}{\partial y} \right) \quad (6)$$

ただし、右辺の最終項において、 Δt は時間ステップをすすめる場合の時間刻み幅であり、離散化誤差を考慮して本来は0になるべき括弧内の項は残している。

一般座標

$$\xi = \xi(x, y), \quad \eta = \eta(x, y)$$

を用いた場合、ナビエ・ストークス方程式の非線形項は、x方向成分に対して

$$\begin{aligned} uu_x + vu_y &= u(u_\xi \xi_x + u_\eta \eta_x) + v(u_\xi \xi_y + u_\eta \eta_y) \\ &= (u_\xi x + v_\xi y)u_\xi + (u_\eta x + v_\eta y)u_\eta \end{aligned} \quad (7)$$

となるため、(U,V)を

$$U = u_\xi x + v_\xi y = u \frac{\partial \xi}{\partial x} + v \frac{\partial \xi}{\partial y} \quad (8)$$

$$V = u_\eta x + v_\eta y = u \frac{\partial \eta}{\partial x} + v \frac{\partial \eta}{\partial y} \quad (9)$$

で定義すれば

$$uu_x + vu_y = Uu_\xi + Vu_\eta \quad (10)$$

と書ける。このときナビエ・ストークス方程式(4)、(5)は

$$\frac{\partial u}{\partial t} = -U \frac{\partial u}{\partial \xi} - V \frac{\partial u}{\partial \eta} - \left(\frac{\partial \xi}{\partial x} \frac{\partial p}{\partial \xi} + \frac{\partial \eta}{\partial x} \frac{\partial p}{\partial \eta} \right) + \frac{1}{Re} \Delta u \quad (11)$$

$$\frac{\partial v}{\partial t} = -U \frac{\partial v}{\partial \xi} - V \frac{\partial v}{\partial \eta} - \left(\frac{\partial \xi}{\partial y} \frac{\partial p}{\partial \xi} + \frac{\partial \eta}{\partial y} \frac{\partial p}{\partial \eta} \right) + \frac{1}{Re} \Delta v \quad (12)$$

となる。圧力のポアソン方程式は次のように変換される。

$$\begin{aligned} \Delta p = & -(u_\xi \xi_x + u_\eta \eta_x)^2 - 2(u_\xi \xi_y + u_\eta \eta_y)(v_\xi \xi_x + v_\eta \eta_x) - (v_\xi \xi_y + v_\eta \eta_y)^2 \\ & + (u_\xi \xi_x + u_\eta \eta_x + v_\xi \xi_y + v_\eta \eta_y) / \Delta t \end{aligned} \quad (13)$$

$$\xi_x = \frac{\partial \xi}{\partial x} = \frac{1}{J} \frac{\partial y}{\partial \eta}, \quad \eta_y = \frac{\partial \eta}{\partial x} = -\frac{1}{J} \frac{\partial y}{\partial \xi} \quad (14)$$

$$\xi_y = \frac{\partial \xi}{\partial y} = -\frac{1}{J} \frac{\partial x}{\partial \eta}, \quad \eta_x = \frac{\partial \eta}{\partial y} = \frac{1}{J} \frac{\partial x}{\partial \xi} \quad (15)$$

ただし、

$$J = \frac{\partial x}{\partial \xi} \frac{\partial y}{\partial \eta} - \frac{\partial x}{\partial \eta} \frac{\partial y}{\partial \xi} \quad (16)$$

また、 Δ は一般座標でのラプラシアンであり、具体的には

$$\begin{aligned} \Delta f = & C_1 f_{\xi\xi} + C_2 f_{\eta\eta} + C_3 f_{\xi\eta} + C_4 f_\xi + C_5 f_\eta \\ C_1 = & \xi_x^2 + \xi_y^2, \quad C_2 = \eta_x^2 + \eta_y^2, \quad C_3 = 2(\xi_x \eta_x + \xi_y \eta_y) \\ C_4 = & \xi_{xx} + \xi_{yy} = \xi_x(\xi_x)_\xi + \eta_x(\xi_x)_\eta + \xi_y(\xi_y)_\xi + \eta_y(\xi_y)_\eta \\ C_5 = & \eta_{xx} + \eta_{yy} = \xi_x(\eta_x)_\xi + \eta_x(\eta_x)_\eta + \xi_y(\eta_y)_\xi + \eta_y(\eta_y)_\eta \end{aligned} \quad (17)$$

である。なお、 C_4 、 C_5 内の $(\xi_x)_\xi$ などは、上式をもう一度 ξ 、 η で微分すれば式の形で表現することもできるが、非常に煩雑である。そこで、計算の過程で ξ_x を(配列に)記憶しておき、必要に応じて

$$(\xi_x)_\xi = \{(\xi_x)_{i+1,j,k} - (\xi_x)_{i-1,j,k}\} / 2\Delta\xi$$

などから計算した方が簡単である。

本研究では、差分法を用いて方程式(11)、(12)、(13)を解いた。すべての空間微分は2次精度中心差分で近似し、時間微分は1次精度の前進差分

$$\frac{\partial u}{\partial t} = \lim_{\Delta t \rightarrow 0} \frac{u(t + \Delta t) - u(t)}{\Delta t} \sim \frac{u^{n+1} - u^n}{\Delta t} \quad (18)$$

によって近似した。ある時間ステップ n での速度を用いて、式(13)の右辺を計算し、対応する圧力 p はポアソン方程式を、SOR法を用いて反復的に解いて求める。次に得られた圧力を用いて式(11)、(12)の近似式

$$u^{n+1} = u^n + \Delta t f(u^n, v^n, p) \quad (19)$$

$$v^{n+1} = v^n + \Delta t g(u^n, v^n, p) \quad (20)$$

から次の時間ステップ $n+1$ での速度を求める。ここで f と g はそれぞれ式 (11)、(12) の右辺である。

初期条件は一様流れとした。境界条件は円柱上で速度が0（粘着条件）、遠方で一様流とした。通常の流れでは 2π 周期の周期条件を課すが、2枚のリーマン面の場合は 4π 周期、3枚のリーマン面の場合に 6π 周期の周期条件を課した。

楕円柱は長軸と短軸の比が3：4の縦長の楕円柱として、周方向に1枚の面に対して60格子をとり、半径方向には楕円柱から遠ざかるにつれて幅が広がる35格子を用いた。図3、4に本研究で用いた格子（全体と楕円柱近傍）を示す。




図3 格子図（全領域）




図4 格子図（楕円柱近傍）

4. 計算結果と考察

以下にレイノルズ数が200の場合について、無次元の時間刻み幅 Δt を0.01とし、3000ステップにおける計算結果を示す。流れは x の負の方向から正の方向に向かって流れている。

図5には、リーマン面が1枚、すなわち現実に起きる流れを速度ベクトルで表示したものである。流れは楕円柱の短軸に対して 0° （迎角 0° ）であたっている。楕円柱背後にカルマン渦が見られる。また図6は同じ時間ステップでの等圧線図である。図5で見られた渦の中心付近の圧力が低くなっていることがわかる。




図5 楕円柱まわりの流れの速度ベクトル




図6 楕円柱まわりの流れの等圧線図

リーマン面が複数枚の場合は、原点から x 軸の正の部分に沿って切れ目を入れて各面をつないでいる。すなわち、 x 軸

から測った角度が $0^\circ \sim 360^\circ$ となるものを1枚目のリーマン面（面1）、 $360^\circ \sim 720^\circ$ になるものを2枚目（面2、以下同様）とする。楕円柱の後流部分に着目する場合は、x軸を挟んで流れが不連続に見えないようにするため、角度が $0^\circ \sim 180^\circ$ と $540^\circ \sim 720^\circ$ の面（面A）および $180^\circ \sim 540^\circ$ の面（面B）を表示する。リーマン面が3枚の場合は、角度が $0^\circ \sim 180^\circ$ と $900^\circ \sim 1080^\circ$ の面（面A）および $180^\circ \sim 540^\circ$ の面（面B）そして $540^\circ \sim 900^\circ$ の面（面C）を表示する。図7、8は、リーマン面が2枚の場合で面1において迎角が 1° 、面2において迎角が 0° の場合の計算結果を速度ベクトルで表示したもので、図7は面A、図8は面Bに対応する。




図7 角度が $0^\circ \sim 180^\circ$ と $540^\circ \sim 720^\circ$ の面




図8 角度が $180^\circ \sim 540^\circ$ の面

どちらの面でもカルマン渦の放出が見られるが、図9のように2つの図を重ねてみると2つの図の速度ベクトルには差があることがわかる。これは遠方の速度の向き（迎角）が 1° ずれていることに対応して、渦の放出位置がずれたことが原因と考えられる。




図9 楕円柱の近傍で図7と図8を重ねた図

図10、11、12はリーマン面が3枚の場合の計算結果を速度ベクトルで表示した図である。面1の迎角は 1° 、面2の迎角は 0° 、面3の迎角は -1° である。面Aではカルマン渦が見られず、下流側に伸びたやや非対称な双子渦がみられる。面Bと面Cではカルマン渦の放出が見られるが、面Bの速度ベクトルをx軸に関して折り返したものがほぼ面Cの速度ベクトルになっていることから、放出の位相がおよそ 180° ずれていると予想される。各リーマン面の迎角の違いによる剥離点の差および各リーマン面間の相互作用が影響していると考えられる。




図 10 角度が $0^\circ \sim 180^\circ$ と $90^\circ \sim 1080^\circ$ の面




図 11 角度が $180^\circ \sim 540^\circ$ の面




図 12 角度が $540^\circ \sim 900^\circ$ の面

図 13、14 は 2 枚および 3 枚のリーマン面ですべて迎角を 0° にしたときの計算結果である。この場合、面 A と面 B では（3 枚の場合は面 C についても）ほぼ同じ速度分布で、カルマン渦の放出はなく、通常より太く、やや非対称な比較的下流側に伸びた双子渦が見られる。図 5 で見られたカルマン渦の放出が抑えられたのは、各リーマン面間の流れの相互作用が、流れの非対称性を抑える形で働いたものと予測できる。




図 13 2 枚のリーマン面（迎角 0° ）




図 14 3 枚のリーマン面（迎角 0° ）

5. まとめ

本研究では、どのような状況もつくり出せるという数値シミュレーションの利点を活かして、リーマン面内の流体の仮想的な流れのシミュレーションを行った。具体的には2枚のリーマン面と3枚のリーマン面内に楕円柱が置かれて、それに一様流があたっている状態を想定して流れのシミュレーションを行った。基礎方程式として非圧縮性のナビエ・ストークス方程式を用い、標準的なMAC法を用いて方程式を数値的に解いた。レイノルズ数はカルマン渦が発生する200として、現実におきる流れと比較した。リーマン面が2枚の場合、1枚目と2枚目の迎角を 1° と0度にした場合にはカルマン渦が発生し、その位相はリーマン面ごとにわずかな差がみられた。リーマン面が3枚の場合で迎角を 1° 、 0° 、 -1° とした場合、リーマン面によって双子渦の流れと位相がほぼ 180° ずれたカルマン渦の流れが見られた。各リーマン面で迎角をすべて 0° にした場合、各面の流れはほぼ同じで、双子渦は生じたものの、カルマン渦までは発達しなかった。

参考文献

- 1 Tetuya Kawamura: Numerical simulation of the flow around a four dimensional sphere in a four dimensional duct, Natural Science Report, Ochanomizu University, Vol.72, No.2 (2022)
- 2 河村哲也：流体解析の基礎，朝倉書店，ISBN978-4-254-13111 (2014)

付録 プログラム

本研究で用いたプログラムを以下に示す。Fortran で記述されている。1-29 行目がメインプログラムであり、14 行目の NSTEPS が計算終了までの時間ステップ数である。18-22 行目が圧力のポアソン方程式を SOR 法で解く部分であり、18 行目の ISTEP0 は最大反復回数、21 行目で反復法の収束判定を行っており、すべての格子点での反復前後の差の2乗和 ERR が判定値 ERR より小さくなった時点で収束したとみなしている。メインプログラムに用いているサブルーチンプログラムの意味は以下のとおりである。

サブルーチン GRID では楕円柱まわりの格子を生成している。NPR がリーマン面の枚数でプログラム実行時に入力する形になっている。JMAX は周方向の格子数でリーマン面1枚につき60格子をとっている。KMAX は径方向の格子数で35格子をとっている。それぞれ増やすことは可能であるが、そのときはそれに合わせて各プログラムにあるパラメータ文の JDIM と KDIM の数値を増やす必要がある。AA と BB が楕円柱の短径および長径の長さ、HH は楕円柱から1格子目の径方向の格子幅である。この格子幅が外側に向かって等比級数的に RA 倍ずつ拡大されて広がる。なお、別のプログラムで生成された格子を取り込むことも可能である。

サブルーチン DATA では計算に用いる時間刻み幅やレイノルズ数、ポアソン方程式の反復回数などパラメータを入力する。プログラム実行時に入力する形で、標準値が画面に表示される。計算方法（空間方向に中心差分を用いる方法）の関係でレイノルズ数が200程度より小さくとらないと計算が発散する。またレイノルズ数が小さいと、それに合わせて時間刻み幅を小さくとる必要がある。遠方で強制的に一様流になるという境界条件を課しているため、流れが遠方境界に達する程度に長時間計算すると計算が不安定になる可能性がある。

サブルーチン METJ は一般座標を用いたために方程式に現れる ξ x などのメトリックを格子データから計算している。配列 XX,XY,YX,YY はそれぞれ ξ x, ξ y, η x, η y であり、AJ は式 (16) で定義される変換のヤコビアン、C1 ~ C5 はラプラシアンを計算するときに現れる係数であり、式 (18) に対応させている。

サブルーチン INIT は初期条件であり、速度は角度 ALP 度で楕円柱にあたる一様流、圧力は 0 としている。リーマン面が 2 枚の場合は、2 枚目は 0 であたるとし、リーマン面が 3 枚の場合は、2 枚目は角度 0、3 枚目は角度 $-\text{ALP}$ 度であたるとしている。サブルーチン OUT は結果の出力部分で、50 ステップごとに速度と圧力をファイルに出力している。

サブルーチン RHS はある時間ステップでの速度を用いてポアソン方程式の右辺を計算する部分、サブルーチン PRS は反復法の 1 回の反復で各格子点での圧力を変化させる部分、サブルーチン PRSBC は圧力の境界条件である。この場合、円柱表面と遠方境界では面と交差する方向に圧力差はないという近似的な条件を用いている。なお、周方向には周期条件（リーマン面が 1 枚のときは 2π 周期、2 枚のときは 4π 周期、3 枚のときは 6π 周期）を与えている。

サブルーチン NS は式 (19)、(20) を用いて、ある時間ステップの速度から次の時間ステップでの速度を計算する部分である。320 行目と 321 行目の UA と VA は式 (8)、(9) に対応する。また 322 行目は式 (10) で表される x 方向速度 U の非線形項であり 323 行目は y 方向速度 V の非線形項である。324 行目 UVS は x 方向速度 U のラプラシアン、328 行目の VVS は y 方向速度のラプラシアンである。333 行目と 335 行目で次の時間ステップでの U と V の値をいったん配列 D と Q に記憶しておいて、各格子点での値が修正された時点で、340 行目と 341 行目で D と Q の値を次の時間ステップの U と V の値として記憶している。サブルーチン BC は速度の境界条件であり、遠方では初期条件と同じ一様流、楕円柱上で速度 0 にしている。なお、周方向は圧力と同じ周期条件を与えている。

河村哲也 湖国カルチャーセンター客員教授

```

1 C*****
2 C    2 dimensional flow slover (genaralized coordinate sysytem) *
3 C*****
4 PROGRAM FLOW2D
5 PARAMETER (JDIM=182, KDIM=35)
6 COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEP0, NSTEPS, ITYP, IGR, ALP, NPR
7 COMMON/FLWPRM/ RE, REI, DT, DTI, EPS, ERR, CONST
8 C
9 CALL GRID
10 CALL DATA
11 CALL METJ
12 CALL INIT
13 C
14 DO 10 N = 0, NSTEPS
15 IF (MOD(N, 50).EQ.0) CALL OUTP(N, DT*N)
16 CALL RHS
17 ERR = 0.
18 DO 20 I=1, ISTEP0
19 CALL PRS
20 CALL PRSBC
21 IF (ERR.LT.EPS) GO TO 30
22 20 CONTINUE
23 30 WRITE(*,*) N, ERR
24 CALL NS
25 CALL BC
26 10 CONTINUE
27 C
28 STOP
29 END
30 C*****
31 C SUBROUTINE FOR INPUT DATA *
32 C*****
33 SUBROUTINE DATA
34 COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEP0, NSTEPS, ITYP, IGR, ALP, NPR
35 COMMON/FLWPRM/ RE, REI, DT, DTI, EPS, ERR, CONST
36 JM = JMAX-1
37 KM = KMAX-1
38 WRITE(*,*) 'INPUT NUMBER OF TIME STEP (3000)'
39 READ(*,*) NSTEPS
40 WRITE(*,*) 'INPUT MAX. NUMBER OF ITERATION FOR POISSON EQ. (30)'
41 READ(*,*) ISTEP0
42 WRITE(*,*) 'INPUT REYNOLDS NUM. & TIME INCREMENT DT (200., .01)'
43 READ(*,*) RE, DT
44 WRITE(*,*) 'INPUT MAX. ERROR (.0001)'
45 READ(*,*) EPS
46 WRITE(*,*) 'INPUT ACCERARATION PARAMETER FOR SOR METHOD (1)'
47 READ(*,*) CONST
48 REI = 1./RE
49 DTI = 1./DT
50 RETURN
51 END
52 C*****
53 C SUBROUTINE FOR GRID DATA *
54 C*****
55 SUBROUTINE GRID
56 PARAMETER ( JDIM=182, KDIM=35)
57 COMMON/GRIDXY/ X(JDIM, KDIM), Y(JDIM, KDIM)
58 COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEP0, NSTEPS, ITYP, IGR, ALP, NPR
59 DIMENSION RR(KDIM)
60 C
61 C WRITE(*,*) 'DO YOU WANT TO READ GRID FILE ? (YES=1, NO=0) (0)'
62 C READ(*,*) IGR
63 C WRITE(*,*) 'Is the grid Periodic one ? (Yes=1, No=0) (1)'
64 C READ(*,*) ITYP
65 IGR=0
66 ITYP=1
67 WRITE(*,*) 'Number of Riemann-Plane (1 or 2 or 3) ?'
68 READ(*,*) NPR
69 C
70 IF (IGR.EQ.1) THEN
71 C
72 READ(11,*) IA, IB
73 READ(11,*) ((X(I, J), I=1, IA), J=1, IB)
74 READ(11,*) ((Y(I, J), I=1, IA), J=1, IB)
75 JMAX = IA
76 KMAX = IB
77 C
78 ELSE
79 C
80 C*** GRID AROUND THE ELLIPSOID
81 JMAX = 60*NPR+2
82 KMAX = 35
83 PAI = ATAN(1.) *4.
84 AA = 0.75
85 BB = 1.0
86 HH = 0.025

```

```

87      RA = 1.15
88      RR(1) = 1.0
89      DO 15 K = 2, KMAX
90          RR(K) = RR(K-1) + HH*RA** (K-1)
91      15 CONTINUE
92      WRITE(*,*) (RR(K), K=1, KMAX)
93      DO 20 K = 1, KMAX
94          DO 20 J = 1, JMAX
95              TT = 2.*PAI*FLOAT(J-2)/FLOAT(JMAX-2)*NPR
96              BC = BB+(AA-BB)*FLOAT(K-1)/FLOAT(KMAX-1)
97              X(J, K) = AA*RR(K)*COS(TT)
98              Y(J, K) = BC*RR(K)*SIN(TT)
99      20 CONTINUE
100     END IF
101 C
102     RETURN
103     END
104 C
105 C*****
106 C SUBROUTINE FOR CALCULATING METRICS *
107 C*****
108     SUBROUTINE METJ
109     PARAMETER ( JDIM=182, KDIM=35)
110     COMMON/GRIDXY/ X(JDIM, KDIM), Y(JDIM, KDIM)
111     COMMON/METRIC/
112     1 XX(JDIM, KDIM), XY(JDIM, KDIM), YX(JDIM, KDIM), YY(JDIM, KDIM),
113     2 C1(JDIM, KDIM), C2(JDIM, KDIM), C3(JDIM, KDIM),
114     3 C4(JDIM, KDIM), C5(JDIM, KDIM), AJ(JDIM, KDIM)
115     COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEP0, NSTEPS, ITYP, IGR, ALP, NPR
116     DO 125 K=1, KMAX
117         DO 125 J=1, JMAX
118             IF (K.EQ.1) THEN
119                 XE=0.5*(-X(J, 3)+4.0*X(J, 2)-3.0*X(J, 1))
120                 YE=0.5*(-Y(J, 3)+4.0*Y(J, 2)-3.0*Y(J, 1))
121             ELSE IF (K.EQ.KMAX) THEN
122                 XE=0.5*(X(J, KMAX-2)-4.0*X(J, KMAX-1)+3.0*X(J, KMAX))
123                 YE=0.5*(Y(J, KMAX-2)-4.0*Y(J, KMAX-1)+3.0*Y(J, KMAX))
124             ELSE
125                 XE=0.5*(X(J, K+1)-X(J, K-1))
126                 YE=0.5*(Y(J, K+1)-Y(J, K-1))
127             END IF
128             IF (J.EQ.1) THEN
129                 XXI=0.5*(-X(3, K)+4.0*X(2, K)-3.0*X(1, K))
130                 YXI=0.5*(-Y(3, K)+4.0*Y(2, K)-3.0*Y(1, K))
131             IF (ITYP.EQ.1) THEN
132                 XXI=0.5*(X(2, K)-X(JMAX-2, K))
133                 YXI=0.5*(Y(2, K)-Y(JMAX-2, K))
134             END IF
135             ELSE IF (J.EQ.JMAX) THEN
136                 XXI=0.5*(X(JMAX-2, K)-4.0*X(JMAX-1, K)+3.0*X(JMAX, K))
137                 YXI=0.5*(Y(JMAX-2, K)-4.0*Y(JMAX-1, K)+3.0*Y(JMAX, K))
138             IF (ITYP.EQ.1) THEN
139                 XXI=0.5*(X(3, K)-X(JMAX-1, K))
140                 YXI=0.5*(Y(3, K)-Y(JMAX-1, K))
141             END IF
142             ELSE
143                 XXI=0.5*(X(J+1, K)-X(J-1, K))
144                 YXI=0.5*(Y(J+1, K)-Y(J-1, K))
145             END IF
146             IF (ITYP.EQ.1.AND.J.EQ.1) THEN
147                 XXI=0.5*(X(J+1, K)-X(JM-1, K))
148                 YXI=0.5*(Y(J+1, K)-Y(JM-1, K))
149             END IF
150             IF (ITYP.EQ.1.AND.J.EQ.JMAX) THEN
151                 XXI=0.5*(X(3, K)-X(J-1, K))
152                 YXI=0.5*(Y(3, K)-Y(J-1, K))
153             END IF
154             AJJ=XXI*YE-XE*YXI
155             XX(J, K)=YE/AJJ
156             YX(J, K)=-YXI/AJJ
157             XY(J, K)=-XE/AJJ
158             YY(J, K)=XXI/AJJ
159             AJ(J, K)=AJJ
160      125 CONTINUE
161      DO 10 K=1, KMAX
162          DO 10 J=1, JMAX
163              C1(J, K)=XX(J, K)**2+XY(J, K)**2
164              C3(J, K)=YX(J, K)**2+YY(J, K)**2
165              C2(J, K)=2.*(XX(J, K)*YX(J, K)+XY(J, K)*YY(J, K))
166      10 CONTINUE
167      DO 11 K=2, KM
168          DO 11 J=2, JM
169              C77=XX(J, K)*(XX(J+1, K)-XX(J-1, K))+YX(J, K)*(XX(J, K+1)
170      1 -XX(J, K-1))+XY(J, K)*(XY(J+1, K)-XY(J-1, K))
171      2 +YY(J, K)*(XY(J, K+1)-XY(J, K-1))
172              C88=XX(J, K)*(YX(J+1, K)-YX(J-1, K))+YX(J, K)*(YX(J, K+1)

```

```

173      1  -YX(J,K)+XY(J,K)*(YY(J+1,K)-YY(J-1,K))
174      2  +YY(J,K)*(YY(J,K+1)-YY(J,K-1))
175      C4(J,K)=C77*.5
176      C5(J,K)=C88*.5
177      11 CONTINUE
178      RETURN
179      END
180 C*****
181 C  SUBROUTINE FOR INITIAL CONDITION *
182 C*****
183      SUBROUTINE INIT
184      PARAMETER ( JDIM=182, KDIM=35)
185      COMMON/GRIDXY/ X(JDIM, KDIM), Y(JDIM, KDIM)
186      COMMON/METRIC/
187      1 XX(JDIM, KDIM), XY(JDIM, KDIM), YX(JDIM, KDIM), YY(JDIM, KDIM),
188      2 C1(JDIM, KDIM), C2(JDIM, KDIM), C3(JDIM, KDIM),
189      3 C4(JDIM, KDIM), C5(JDIM, KDIM), AJ(JDIM, KDIM)
190      COMMON /VEL/ U(JDIM, KDIM), V(JDIM, KDIM), P(JDIM, KDIM)
191      COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEPO, NSTEPS, ITYP, IGR, ALP, NPR
192      WRITE(*,*) ' INPUT ATTACK ANGLE? (1)'
193      READ(*,*) ALP
194      TTT = ALP*3.1415192/180.
195      DO 20 K = 1, KMAX
196      DO 10 J = 1, JMAX
197          U(J, K) = COS(TTT)
198          V(J, K) = SIN(TTT)
199          P(J, K) = 0.
200      10 CONTINUE
201      IF (NPR.EQ.2) THEN
202      DO 15 J = JMAX/2+1, JMAX
203          U(J, K) = COS(TTT*0.)
204          V(J, K) = SIN(TTT*0.)
205      15 CONTINUE
206      END IF
207      IF (NPR.EQ.3) THEN
208      DO 11 J = JMAX/3+1, 2*JMAX/3
209          U(J, K) = COS(TTT*0.)
210          V(J, K) = SIN(TTT*0.)
211      11 CONTINUE
212      DO 12 J = 2*JMAX/3+1, jmax
213          U(J, K) = COS(-TTT)
214          V(J, K) = SIN(-TTT)
215      12 CONTINUE
216      END IF
217      20 CONTINUE
218      RETURN
219      END
220 C*****
221 C  SUBROUTINE FOR CALCULATING RHS OF POISSON EQ. *
222 C*****
223      SUBROUTINE RHS
224      PARAMETER ( JDIM=182, KDIM=35)
225      COMMON /VEL/ U(JDIM, KDIM), V(JDIM, KDIM), P(JDIM, KDIM)
226      COMMON /FLW/ Q(JDIM, KDIM), D(JDIM, KDIM)
227      COMMON/METRIC/
228      1 XX(JDIM, KDIM), XY(JDIM, KDIM), YX(JDIM, KDIM), YY(JDIM, KDIM),
229      2 C1(JDIM, KDIM), C2(JDIM, KDIM), C3(JDIM, KDIM),
230      3 C4(JDIM, KDIM), C5(JDIM, KDIM), AJ(JDIM, KDIM)
231      COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEPO, NSTEPS, ITYP, IGR, ALP, NPR
232      COMMON/FLWPRM/ RE, REI, DT, DTI, EPS, ERR, CONST
233      DO 10 K = 2, KM
234      DO 10 J = 2, JM
235          UXD = XX(J, K)*(U(J+1, K)-U(J-1, K))+YX(J, K)*(U(J, K+1)-U(J, K-1))
236          UYD = XY(J, K)*(U(J+1, K)-U(J-1, K))+YY(J, K)*(U(J, K+1)-U(J, K-1))
237          VXD = XX(J, K)*(V(J+1, K)-V(J-1, K))+YX(J, K)*(V(J, K+1)-V(J, K-1))
238          VYD = XY(J, K)*(V(J+1, K)-V(J-1, K))+YY(J, K)*(V(J, K+1)-V(J, K-1))
239          Q(J, K) = -.25*(UXD*UXD+2.*UYD*VXD+VYD*VYD)+.5*(UXD+VYD)*DTI
240      10 CONTINUE
241      RETURN
242      END
243 C*****
244 C  SUBROUTINE FOR PRESSURE BOUNDARY CONDITION *
245 C*****
246      SUBROUTINE PRSBC
247      PARAMETER ( JDIM=182, KDIM=35)
248      COMMON /VEL/ U(JDIM, KDIM), V(JDIM, KDIM), P(JDIM, KDIM)
249      COMMON/METRIC/
250      1 XX(JDIM, KDIM), XY(JDIM, KDIM), YX(JDIM, KDIM), YY(JDIM, KDIM),
251      2 C1(JDIM, KDIM), C2(JDIM, KDIM), C3(JDIM, KDIM),
252      3 C4(JDIM, KDIM), C5(JDIM, KDIM), AJ(JDIM, KDIM)
253      COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEPO, NSTEPS, ITYP, IGR, ALP, NPR
254      COMMON/FLWPRM/ RE, REI, DT, DTI, EPS, ERR, CONST
255 C*** no-slip at K=1 free stream at K=KMAX
256      DO 10 J = 2, JM
257          UL = .5*C2(J, 1)*(U(J+1, 2)-U(J-1, 2))+C5(J, 1)*U(J, 1)
258          VL = 2.*C3(J, 1)*V(J, 2)

```

```

259      P(J,1) = P(J,2)-REI*AJ(J,1)*(XX(J,1)*VL-XY(J,2)*UL)
260      P(J,KMAX) = -.5*(U(J,KMAX)**2+V(J,KMAX)**2)
261 10 CONTINUE
262 C
263      IF(ITYP.EQ.1) THEN
264 C*** periodic
265          DO 20 K = 1, KMAX
266              P(1,K) = P(JM,K)
267              P(JMAX,K) = P(2,K)
268          20 CONTINUE
269      ELSE
270 C*** extrapolate
271          DO 30 K = 1, KMAX
272              P(1,K) = P(2,K)
273              P(JMAX,K) = P(JM,K)
274          30 CONTINUE
275      END IF
276      RETURN
277      END
278 C*****
279 C SUBROUTINE FOR CALCULATING PRESSURE *
280 C*****
281 SUBROUTINE PRS
282     PARAMETER ( JDIM=182, KDIM=35)
283     COMMON /VEL/ U(JDIM, KDIM), V(JDIM, KDIM), P(JDIM, KDIM)
284     COMMON /FLW/ Q(JDIM, KDIM), D(JDIM, KDIM)
285     COMMON/METRIC/
286     1 XX(JDIM, KDIM), XY(JDIM, KDIM), YX(JDIM, KDIM), YY(JDIM, KDIM),
287     2 C1(JDIM, KDIM), C2(JDIM, KDIM), C3(JDIM, KDIM),
288     3 C4(JDIM, KDIM), C5(JDIM, KDIM), AJ(JDIM, KDIM)
289     COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEP0, NSTEPS, ITYP, IGR, ALP, NPR
290     COMMON/FLWPRM/ RE, REI, DT, DTI, EPS, ERR, CONST
291     DO 10 K = 2, KM
292     DO 10 J = 2, JM
293         CC = .5/(C1(J,K)+C3(J,K))
294         PA = C1(J,K)*(P(J+1,K)+P(J-1,K))+C3(J,K)*(P(J,K+1)+P(J,K-1))
295             +.25*C2(J,K)*(P(J+1,K+1)-P(J-1,K+1)-P(J+1,K-1)+P(J-1,K-1))
296             +.5*C4(J,K)*(P(J+1,K)-P(J-1,K))
297             +.5*C5(J,K)*(P(J,K+1)-P(J,K-1))
298         PP = (PA-Q(J,K))*CC
299         ERR = ERR+(PP-P(J,K))**2
300         P(J,K) = P(J,K)*(1.-CONST)+PP*CONST
301 10 CONTINUE
302     RETURN
303     END
304 C*****
305 C SUBROUTINE FOR CALCULATING VELOCITY *
306 C*****
307 SUBROUTINE NS
308     PARAMETER ( JDIM=182, KDIM=35)
309     COMMON /VEL/ U(JDIM, KDIM), V(JDIM, KDIM), P(JDIM, KDIM)
310     COMMON /FLW/ Q(JDIM, KDIM), D(JDIM, KDIM)
311     COMMON/METRIC/
312     1 XX(JDIM, KDIM), XY(JDIM, KDIM), YX(JDIM, KDIM), YY(JDIM, KDIM),
313     2 C1(JDIM, KDIM), C2(JDIM, KDIM), C3(JDIM, KDIM),
314     3 C4(JDIM, KDIM), C5(JDIM, KDIM), AJ(JDIM, KDIM)
315     COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEP0, NSTEPS, ITYP, IGR, ALP, NPR
316     COMMON/FLWPRM/ RE, REI, DT, DTI, EPS, ERR, CONST
317 C
318     DO 10 K = 2, KM
319     DO 10 J = 2, JM
320         UA = (U(J,K)*XX(J,K)+V(J,K)*XY(J,K))* .5
321         VA = (U(J,K)*YX(J,K)+V(J,K)*YY(J,K))* .5
322         UNL = UA*(U(J+1,K)-U(J-1,K))+VA*(U(J,K+1)-U(J,K-1))
323         VNL = UA*(V(J+1,K)-V(J-1,K))+VA*(V(J,K+1)-V(J,K-1))
324         UVS = C1(J,K)*(U(J+1,K)-2.*U(J,K)+U(J-1,K))
325             +C2(J,K)*(U(J+1,K+1)-U(J+1,K-1)-U(J-1,K+1)+U(J-1,K-1))* .25
326             +C3(J,K)*(U(J,K+1)-2.*U(J,K)+U(J,K-1))+.5*C4(J,K)
327             *(U(J+1,K)-U(J-1,K))+.5*C5(J,K)*(U(J,K+1)-U(J,K-1))
328         VVS = C1(J,K)*(V(J+1,K)-2.*V(J,K)+V(J-1,K))
329             +C2(J,K)*(V(J+1,K+1)-V(J+1,K-1)-V(J-1,K+1)+V(J-1,K-1))* .25
330             +C3(J,K)*(V(J,K+1)-2.*V(J,K)+V(J,K-1))+.5*C4(J,K)
331             *(V(J+1,K)-V(J-1,K))+.5*C5(J,K)*(V(J,K+1)-V(J,K-1))
332 C
333         D(J,K) = U(J,K)+DT*(-UNL-.5*XX(J,K)*(P(J+1,K)-P(J-1,K))
334             -.5*YX(J,K)*(P(J,K+1)-P(J,K-1))+REI*UVS)
335         Q(J,K) = V(J,K)+DT*(-VNL-.5*XY(J,K)*(P(J+1,K)-P(J-1,K))
336             -.5*YY(J,K)*(P(J,K+1)-P(J,K-1))+REI*VVS)
337 10 CONTINUE
338     DO 20 K = 2, KM
339     DO 20 J = 2, JM
340         U(J,K) = D(J,K)
341         V(J,K) = Q(J,K)
342 20 CONTINUE
343     RETURN
344     END

```

```

345 C*****
346 C SUBROUTINE FOR BOUNDARY CONDITION FOR VELOCITY *
347 C*****
348 SUBROUTINE BC
349 PARAMETER ( JDIM=182, KDIM=35)
350 COMMON /VEL/ U(JDIM, KDIM), V(JDIM, KDIM), P(JDIM, KDIM)
351 COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEPO, NSTEPS, ITYP, IGR, ALP, NPR
352 C
353 C*** no-slip at K=1 free stream at K=KMAX
354 TTT=ALP*3.141592/180.
355 DO 10 J = 1, JMAX
356 U(J, 1) = 0.
357 V(J, 1) = 0.
358 U(J, KMAX) = COS(TTT)
359 V(J, KMAX) = SIN(TTT)
360 10 CONTINUE
361 IF (NPR.EQ. 2) THEN
362 DO 15 J = JMAX/2+1, JMAX
363 U(J, KMAX) = COS(TTT*0.)
364 V(J, KMAX) = SIN(TTT*0.)
365 15 CONTINUE
366 END IF
367 IF (NPR.EQ. 3) THEN
368 DO 11 J = JMAX/3+1, 2*JMAX/3
369 U(J, KMAX) = COS(TTT*0.)
370 V(J, KMAX) = SIN(TTT*0.)
371 11 CONTINUE
372 DO 12 J = 2*JMAX/3+1, jmax
373 U(J, KMAX) = COS(-TTT)
374 V(J, KMAX) = SIN(-TTT)
375 12 CONTINUE
376 END IF
377 C
378 IF (ITYP.EQ. 1) THEN
379 C*** periodic
380 DO 20 K = 1, KMAX
381 U(1, K) = U(JM, K)
382 V(1, K) = V(JM, K)
383 U(JMAX, K) = U(2, K)
384 V(JMAX, K) = V(2, K)
385 20 CONTINUE
386 ELSE
387 C*** symmetric
388 DO 30 K = 1, KMAX
389 U(1, K) = U(2, K)
390 V(1, K) = 0.
391 U(JMAX, K) = U(JM, K)
392 V(JMAX, K) = 0.
393 30 CONTINUE
394 END IF
395 RETURN
396 END
397 C*****
398 C WRITE DATA INTO FILE *
399 C*****
400 SUBROUTINE OUTP(NN, TT)
401 PARAMETER ( JDIM=182, KDIM=35)
402 COMMON/GRIDXY/ X(JDIM, KDIM), Y(JDIM, KDIM)
403 COMMON /VEL/ U(JDIM, KDIM), V(JDIM, KDIM), P(JDIM, KDIM)
404 COMMON/GRDPRM/ JM, JMAX, KM, KMAX, ISTEPO, NSTEPS, ITYP, IGR, ALP, NPR
405 LMAX=1
406 WRITE(11) JMAX, KMAX, LMAX
407 WRITE(11) (((X(J, K), J=1, JMAX), K=1, KMAX), L=1, LMAX)
408 WRITE(11) (((Y(J, K), J=1, JMAX), K=1, KMAX), L=1, LMAX)
409 WRITE(11) (((O, O, J=1, JMAX), K=1, KMAX), L=1, LMAX)
410 WRITE(12) NN, TT
411 WRITE(12) (((U(J, K), J=1, JMAX), K=1, KMAX), L=1, LMAX)
412 WRITE(12) (((V(J, K), J=1, JMAX), K=1, KMAX), L=1, LMAX)
413 WRITE(12) (((O, O, J=1, JMAX), K=1, KMAX), L=1, LMAX)
414 WRITE(12) (((P(J, K), J=1, JMAX), K=1, KMAX), L=1, LMAX)
415 RETURN
416 END
417

```